

# Functional Programming in *Mathematica*

A4M36TPJ, WS 14/15, Week 7 - Lecture

Zdeněk Buk, [bukz1@fel.cvut.cz](mailto:bukz1@fel.cvut.cz)

Dept. of Computer Science and Engineering

Faculty of Electrical Engineering

Czech Technical University in Prague

Last update: Nov 2014

# Introduction

## Functional Programming

*From Wikipedia, the free encyclopedia:* Functional programming is a programming paradigm, a style of building the structure and elements of computer programs, that treats computation as the evaluation of mathematical functions and **avoids changing-state** and mutable data. It is a **declarative programming paradigm**, which means programming is done with expressions. In functional code, the **output value of a function depends only on the arguments** that are input to the function, so calling a function  $f$  twice with the same value for an argument  $x$  will produce the same result  $f(x)$  each time. **Eliminating side effects**, i.e. changes in state that do not depend on the function inputs, can make it much easier to understand and **predict the behavior of a program**, which is one of the key motivations for the development of functional programming.

- treats computation as the evaluation of mathematical functions
- avoids changing-state
- output value of a function depends only on the arguments
- eliminating side effects can make it much easier to predict the behavior of a program

## Functional Programming Concepts

### Higher-Order Functions

- functions that can either take other functions as arguments or return them as results

In[1]:= `f[g_, x_] := g[x]`

In[2]:= `f[Sqrt, x]`

Out[2]=  $\sqrt{x}$

In[3]:= `f[g_, x_] := Function[{y}, g[x + y]]`

In[4]:= `f[Sqrt, 42]`

Out[4]= `Function[{y$},  $\sqrt{42 + y\$}$ ]`

In[5]:= `f[Sqrt, 42][z]`


Out[5]=  $\sqrt{42 + z}$

### Pure Functions

Purely functional functions have no side effects.

- If the result of a pure expression is not used, it can be removed without affecting other expressions.
- If there is no data dependency between two pure expressions, then their order can be reversed, or they can be performed in parallel.
- If a pure function is called with arguments that cause no side-effects, the result is constant with respect to that argument list.

In[6]:= **tutorial/PureFunctions**

Out[6]= 

In[7]:= **Function**[{**x**}, **x**<sup>2</sup> + 1]

Out[7]= **Function**[{**x**}, **x**<sup>2</sup> + 1]

In[8]:= **Function**[{**x**}, **x**<sup>2</sup> + 1][**z**]

Out[8]= 1 + **z**<sup>2</sup>

In[9]:= **#**<sup>2</sup> + 1 & [**z**]

Out[9]= 1 + **z**<sup>2</sup>

In[10]:= **Map**[**#**<sup>2</sup> + 1 &, {1, 2, 3, 4, 5, 6}]

Out[10]= {2, 5, 10, 17, 26, 37}

In[11]:= **ParallelMap**[**#**<sup>2</sup> + 1 &, {1, 2, 3, 4, 5, 6}]

Out[11]= {2, 5, 10, 17, 26, 37}

## Recursion

Iteration (looping) in functional languages is usually accomplished via recursion.

In[12]:= **Clear**[**f**];  
**f**[**x\_**] := 1 /; **x** < 1  
**f**[**x\_**] := **x** \* **f**[**x** - 1]

In[15]:= **f**[1]

Out[15]= 1

In[16]:= **f**[4]

Out[16]= 24

In[17]:= **f**[10]

Out[17]= 3 628 800

In[18]:= **Clear**[**g**];  
**Nest**[**g**, **x**, 3]

Out[19]= **g**[**g**[**g**[**x**]]]

In[20]:= **Clear**[**f**, **x**, **a**, **b**, **c**, **d**, **e**]

In[21]:= **Fold**[**f**, **x**, {**a**, **b**, **c**, **d**}]

Out[21]= **f**[**f**[**f**[**f**[**x**, **a**], **b**], **c**], **d**]

---

# Functions in *Mathematica*

## Defining Functions

```
In[22]:= Clear[f, g];  
f[x_] := g[x] + 1
```

```
In[24]:= f[10]
```

```
Out[24]= 1 + g[10]
```

```
In[25]:= Clear[f];  
f = Function[{x}, g[x] + 1]
```

```
Out[26]= Function[{x}, g[x] + 1]
```

```
In[27]:= f[10]
```

```
Out[27]= 1 + g[10]
```

```
In[28]:= Function[{x}, g[x] + 1][10]
```

```
Out[28]= 1 + g[10]
```

```
In[29]:= g[#] + 1 &[10]
```

```
Out[29]= 1 + g[10]
```

```
In[30]:= g[#x] + 1 &[<|"x" → 10|>]
```

```
Out[30]= 1 + g[10]
```

## Multiple Arguments

```
In[31]:= Clear[f]  
f[x_, y_] := x^2 + y + 1
```

```
In[33]:= f[a, b]
```

```
Out[33]= 1 + a^2 + b
```

```
In[34]:= Function[{x, y}, x^2 + y + 1][a, b]
```

```
Out[34]= 1 + a^2 + b
```

```
In[35]:= #1^2 + #2 + 1 &[a, b]
```

```
Out[35]= 1 + a^2 + b
```

## Calling Function

```
In[36]:= ClearAll[f]
```

```
In[37]:= f[x]
```

```
Out[37]= f[x]
```

In[38]:= **f@x**

Out[38]= **f[x]**

In[39]:= **x // f**

Out[39]= **f[x]**

In[40]:= **f[x\_] := g[x] + 1**

In[41]:= **x // f**

Out[41]= **1 + g[x]**

In[42]:= **x // g[#] + 1 &**

Out[42]= **1 + g[x]**

In[43]:= **data = RandomInteger[{0, 10}, 10]**

Out[43]= **{1, 0, 7, 5, 8, 5, 8, 5, 4, 9}**

In[44]:= **{Min[data], Max[data], Mean[data]}**

Out[44]= **{0, 9,  $\frac{26}{5}$ }**

In[45]:= **ClearAll[f]**

**f[x\_] := {Min[x], Max[x], Mean[x]}**

In[47]:= **f[data]**

Out[47]= **{0, 9,  $\frac{26}{5}$ }**

In[48]:= **{Min[#], Max[#], Mean[#]} &@data**

Out[48]= **{0, 9,  $\frac{26}{5}$ }**

In[49]:= **data // {Min[#], Max[#], Mean[#]} &**

Out[49]= **{0, 9,  $\frac{26}{5}$ }**

In[50]:= **a + b**

Out[50]= **a + b**

In[51]:= **Plus[a, b]**

Out[51]= **a + b**

In[52]:= **a ~Plus~ b**

Out[52]= **a + b**

In[53]:= **{1, 2, 3} ~Join~ {4, 5, 6}**

Out[53]= **{1, 2, 3, 4, 5, 6}**

In[54]:= **a ~ (#1^2 + #2^3 &) ~ b**

Out[54]= **a<sup>2</sup> + b<sup>3</sup>**

## Mapping

```
In[55]:= ClearAll[f];  
         f[x_] := g[x] + 1
```

```
In[57]:= Range[5]
```

```
Out[57]= {1, 2, 3, 4, 5}
```

```
In[58]:= f[Range[5]]
```

```
Out[58]= 1 + g[{1, 2, 3, 4, 5}]
```

```
In[59]:= Map[f, Range[5]]
```

```
Out[59]= {1 + g[1], 1 + g[2], 1 + g[3], 1 + g[4], 1 + g[5]}
```

```
In[60]:= f /@ Range[5]
```

```
Out[60]= {1 + g[1], 1 + g[2], 1 + g[3], 1 + g[4], 1 + g[5]}
```

```
In[61]:= Attributes[f]
```

```
Out[61]= {}
```

```
In[62]:= SetAttributes[f, Listable]
```

```
In[63]:= f[Range[5]]
```

```
Out[63]= {1 + g[1], 1 + g[2], 1 + g[3], 1 + g[4], 1 + g[5]}
```

```
In[64]:= g[x] + 1 & /@ Range[5]
```

```
Out[64]= {1 + g[x], 1 + g[x], 1 + g[x], 1 + g[x], 1 + g[x]}
```

---

# Mathematica: Core Language

## Expressions

```
In[65]:= FullForm[x + y]
```

```
Out[65]/FullForm= Plus[x, y]
```

```
In[66]:= Head[x + y]
```

```
Out[66]= Plus
```

```
In[67]:= FullForm[{x, y, z}]
```

```
Out[67]/FullForm= List[x, y, z]
```

```
In[68]:= Head[{x, y, z}]
```

```
Out[68]= List
```

```
In[69]:= Apply[head, {x, y, z}]
```

```
Out[69]= head[x, y, z]
```

```
In[70]:= Apply[Plus, {x, y, z}]
```

```
Out[70]= x + y + z
```

```
In[71]:= FullForm[Apply[Plus, {x, y, z}]]
```

```
Out[71]/FullForm= Plus[x, y, z]
```

```
In[72]:= Apply[Sequence, {x, y, z}]
```

```
Out[72]= Sequence[x, y, z]
```

```
In[73]:= ClearAll[f];  
f[x___] := g[x]
```

```
In[75]:= f[1, 2, 3, 4]
```

```
Out[75]= g[1, 2, 3, 4]
```

```
In[76]:= ClearAll[f];  
f[x___] := List[x]
```

```
In[78]:= f[1, 2, 3, 4]
```

```
Out[78]= {1, 2, 3, 4}
```

## Own Values

```
In[79]:= a = 42
```

```
Out[79]= 42
```

```
In[80]:= OwnValues[a]
```

```
Out[80]= {HoldPattern[a] => 42}
```

## Down Values

A *DownValue* is defined when the variable itself does not have a meaning, but can get one when combined with the proper arguments. This is the case for the most function definitions.

```
In[81]:= f[x_] := x^2
```

This defines a pattern for  $f$  specifying that each time  $f[...]$  is encountered, it is to be replaced by  $...^2$ . This pattern is meaningless if there is a lonely  $f$ ,

```
In[82]:= f
```

```
Out[82]= f
```

However, when encountered with an argument downwards (i.e. down the internal structure of the command you entered), the pattern applies,

```
In[83]:= f[y]
```

```
Out[83]= y^2
```

```
In[84]:= DownValues[f]
```

```
Out[84]= {HoldPattern[f[x_]] :> x^2, HoldPattern[f[x_]] :> {x}}
```

## Up Values

Sometimes, it's convenient not to associate the rule to the outermost symbol. For example, you may want to have a symbol whose value is 2 when it has a subscript of 1, for example to define a special case in a sequence. This would be entered as follows:

```
In[85]:= c/: Subscript[c, 1] := 2
```

If the symbol  $c$  is encountered, neither of the discussed patterns apply.  $c$  on its own has no own hence no *OwnValue*, and looking down the command tree of  $c$  when seeing *Subscript[c, 1]* yields nothing, since  $c$  is already on an outermost branch. An *UpValue* solves this problem: a symbol having an *UpValue* defines a pattern where not only the children, but also the parents are to be investigated, i.e. Mathematica has to look up the command tree to see whether the pattern is to be applied.

```
In[86]:= UpValues[c]
```

```
Out[86]= {HoldPattern[c_1] :> 2}
```

## Sub Values



# Examples

## Counting

Data in this example are represented as time series of 0's and 1's.

```
In[89]:= SeedRandom[0]
data = RandomInteger[{0, 1}, {20}]
Out[90]= {1, 0, 1, 0, 0, 1, 1, 1, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 1, 0}
```

Task is to get a sum of 0's and 1's for each time step.

## Idea

## Example Solutions

### Recursive Fibonacci Number


```
In[100]:= fib[0] := 0;
fib[1] := 1;
fib[n_] := fib[n - 1] + fib[n - 2] /; n > 1

In[103]:= fib[5]
Out[103]= 5

In[104]:= AbsoluteTiming[fib[27]]
Out[104]= {1.606477, 196418}

In[105]:= AbsoluteTiming[fib[28]]
Out[105]= {2.561646, 317811}

In[106]:= ListLinePlot[AbsoluteTiming[fib[#]] & /@ Range[1, 20]]
Out[106]=
```



```
In[107]:= ClearAll[fib]

In[108]:= fib[0] := 0;
fib[1] := 1;
fib[n_] := (fib[n] = fib[n - 1] + fib[n - 2]) /; n > 1
```

In[111]:= **fib**[5]

Out[111]= 5


In[112]:= **AbsoluteTiming**[**fib**[27]]

Out[112]= {0.000274, 196 418}

In[113]:= **AbsoluteTiming**[**fib**[28]]

Out[113]= {0.000026, 317 811}

In[114]:= **ListLinePlot**[(**AbsoluteTiming**[**fib**[#]] & /@ **Range**[1, 20])][**All, 1**]



## Composition

In[115]:= **model**[**cfg\_List**][**x\_**] := **Plus** @@ **MapIndexed**[#1 **x**^(**First**@#2 - 1) &, **cfg**]

In[116]:= **model**[{10, 20, 30, 40}][**x**]

Out[116]= 10 + 20 **x** + 30 **x**<sup>2</sup> + 40 **x**<sup>3</sup>


In[117]:= **model**[{10, 20, 30, 40}][4]

Out[117]= 3130

In[118]:= **m** = **model**[{2, 3, 0.5}]

Out[118]= **model**[{2, 3, 0.5}]


In[119]:= **Plot**[**m**[**x**], {**x**, -5, 5}]



## Example: LinearModelFit

```
In[120]:= data = {{0, 1}, {1, 0}, {3, 2}, {5, 4}};
```


```
In[121]:= ListPlot[data, PlotTheme -> "Detailed"]
```



```
In[122]:= lm = LinearModelFit[data, x, x]
```

```
Out[122]= FittedModel[0.186441 + 0.694915 x]
```

```
In[123]:= Show[ListPlot[data], Plot[lm[x], {x, 0, 5}], Frame → True]
```



```
In[124]:= FullForm[lm]
```

```
Out[124]/FullForm=
FittedModel[List["Linear", List[0.18644067796610198`, 0.6949152542372881`],
  List[List[x], List[1, x]], List[0, 0]], List[List[1.`, 1.`, 1.`, 1.`]],
  List[List[0, 1], List[1, 0], List[3, 2], List[5, 4]],
  List[List[1.`, 0.`, List[1.`, 1.`, List[1.`, 3.`, List[1.`, 5.`]],
  Function[Null, Internal`LocalizedBlock[List[x], Slot[1]], List[HoldAll]]]
```


In[125]:= **lm["Properties"]**

Out[125]= {AdjustedRSquared, AIC, AICc, ANOVATable, ANOVATableDegreesOfFreedom, ANOVATableEntries, ANOVATableFStatistics, ANOVATableMeanSquares, ANOVATablePValues, ANOVATableSumsOfSquares, BasisFunctions, BetaDifferences, BestFit, BestFitParameters, BIC, CatcherMatrix, CoefficientOfVariation, CookDistances, CorrelationMatrix, CovarianceMatrix, CovarianceRatios, Data, DesignMatrix, DurbinWatsonD, EigenstructureTable, EigenstructureTableEigenvalues, EigenstructureTableEntries, EigenstructureTableIndexes, EigenstructureTablePartitions, EstimatedVariance, FitDifferences, FitResiduals, Function, FVarianceRatios, HatDiagonal, MeanPredictionBands, MeanPredictionConfidenceIntervals, MeanPredictionConfidenceIntervalTable, MeanPredictionConfidenceIntervalTableEntries, MeanPredictionErrors, ParameterConfidenceIntervals, ParameterConfidenceIntervalTable, ParameterConfidenceIntervalTableEntries, ParameterConfidenceRegion, ParameterErrors, ParameterPValues, ParameterTable, ParameterTableEntries, ParameterTStatistics, PartialSumOfSquares, PredictedResponse, Properties, Response, RSquared, SequentialSumOfSquares, SingleDeletionVariances, SinglePredictionBands, SinglePredictionConfidenceIntervals, SinglePredictionConfidenceIntervalTable, SinglePredictionConfidenceIntervalTableEntries, SinglePredictionErrors, StandardizedResiduals, StudentizedResiduals, VarianceInflationFactors}

In[126]:= **lm["FitResiduals"]**

Out[126]= {0.813559, -0.881356, -0.271186, 0.338983}

In[127]:= **ListPlot[lm["FitResiduals"], Filling → Axis]**



---

## References

- <http://mathematica.stackexchange.com/questions/96/what-is-the-distinction-between-downvalues-upvalues-subvalues-and-ownvalues>
- Hudak, Paul: *Conception, evolution, and application of functional programming languages*. September 1989, ACM Computing Surveys 21 (3): 359–411. doi:10.1145/72551.72554; <http://www.dbnet.ece.ntua.gr/~adamo/languages/books/p359-hudak.pdf>